Лекція 8. 
Прийняття рішень із застосуванням логічного програмування


8.1 Реалізація найпростішої діагностичної системи

Ми вже можемо представити найпростішу систему, яка може перебирати факти, порівнювати їх з правилами та визначати деяке рішення. Так наприклад система для визначення типу головного болю
% Правила для діагностики головного болю
diagnose_headache(X) :-
     symptom(X, пульсуючий_біль),
     symptom(X, нудота),
     symptom(X, чутливість_до_світла),
     write('У вас може бути мігрень.').

diagnose_headache(X) :-
     symptom(X, тупий_постійний біль_pain),
     symptom(X, втома),
     symptom(X, труднощі_концентрації),
     write('Ви можете відчувати головний біль напруги.').

diagnose_headache(X) :-
     symptom(X, гострий_біль),
     symptom(X, закладеність_синусів),
     symptom(X, сльозливість),
     write('Можливо, у вас синусит.').

% Припустимо, що є база знань про симптоми у пацієнтів
symptom(john, пульсуючий_біль).
symptom(john, нудота).
symptom(john, чутливість_до_світла).

symptom(анна, тупий_постійний_біль).
symptom(анна, втома).
symptom(анна, труднощі_концентрації).

symptom(mike, гострий_біль).
symptom(mike, закладеність_синусів).
symptom(mike, сльозливість).
?-diagnose_headache(john).

Однак, більш складна система може бути ефективніше побудована за рахунок рекурсивного обходу дерева

8.2. Рекурсивний пошук рішення

Розглянемо два приклади задачі пошуку шляху в лабиринті.
Приклад 1. Нехай заданий лабіринт із дев'яти клітин (див. рис). По світлих клітинах переміщатися можна, по інших немає. 

Необхідно визначити чи існує шлях від входу (точка «1») до виходу (точка «9»). Для вирішення цього завдання опишемо: 
- сукупність можливих кроків по лабіринту як сукупність пар сусідніх
вільних клітин (рис.20): (1,4), (4,5), (5,6), (4,7), (6,9);
– процедуру «шлях», що складається із двох правил:
1) шлях з точки X до точки Y є крок з точки X до точки Y;
2) шлях з точки X до точки Y є крок з точки X до точки Z і шлях з точки Z до точки Y.

Побудуємо програму:
%факти
step(1,4). 
step(4,5).
step(5,6).
step(4,7).
step(6,9).
% правила
way(X,Y):-step(X,Y).
way(X,Y):-step(X,Z),way(Z,Y).
Схема трасування програми пояснює її:

Отже, отримавши мету way(1,9) Пролог намагається її перевірити (див. мітку 1) послідовно переглядаючи факти та правила бази знань. Серед фактів такого не виявляється, а ось серед правил виявляється  навіть більше – є два правила, в заголовку яких стоїть саме предікат way.
Підстановка мети перше в правило змушує Пролог перевіряти його здійсненність. Що відбудеться, якщо знайдеться факт step(1,9). Послідовно зіставляючи його з усіма фактами Пролог переконується, що перше правило із зазначеними аргументами нездійсненне (мітка 2). Тоді Пролог рухається далі за програмою з надією ще кудись прилаштувати початкову мету way(1,9). І стикається з правилом №2 (мітка 3). Воно говорить: шлях (1,9) існує, якщо є крок (1, Y) та шлях (Y, 9). Тобто це правило можна здійснити, якщо здійснені послідовно обидві підціли з тіла речення. Доводиться їх послідовно уніфікувати. Уніфікація першої підцілі step(1,Z) закінчується при аналізі першого ж факту step(1,4) і Z визначається значенням «4» (мітка 4). Після чого відбувається виклик другої підцілі way(4,9). І т.д.
В результаті буде виведене значення true.

Приклад 2. Потрібно знайти шлях, який веде від заданого початкового положення до виходу із лабіринту. Розглянемо приклад плану будинку, наведеного на рис.

 Потрібно знайти вихід із заданої кімнати, наприклад g, назовні.
Задамо фактами двомісне відношення door, яке вказує на кімнати, між якими є двері (door(b, outside) позначає зовнішні двері).
door(b, outside).
door(b, c).
door(b, e).
door(d, c).
door(d, e).
door(f, e).
door(g, e).
door(f, g).

Розглянемо програму із використанням списків. Ми використовуватимемо правило 
path(Start, Finish, Banned, Visited),
 яке дає змогу знайти шлях від початкової до кінцевої кімнати, який не проходить через жодну із кімнат двічі. Для уникнення зациклювань використовується список "заборонених" кімнат Banned
path(Start, Start, _, []) :- !.
path(Start, Finish, Banned, [Start|Visited]) :-
	(door(Start, Room); door(Room, Start)), \+member(Room, Banned),
	path(Room, Finish, [Start| Banned], Visited).
Основне правило знаходження виходу з лабиринту:
exitSearch(Start, Visited) :-path(Start, outside, [], Visited).

Таким чином, запуск програми:
door(b, outside).
door(b, c).
door(b, e).
door(d, c).
door(d, e).
door(f, e).
door(g, e).
door(f, g).

path(Start, Start, _, []) :- !.
path(Start, Finish, Banned, [Start|Visited]) :-(door(Start, Room); door(Room, Start)),\+member(Room, Banned),
 path(Room, Finish, [Start| Banned], Visited).

exitSearch(Start, Visited) :-path(Start, outside, [], Visited).
З запитом:
?- exitSearch(g, R).
Дозволить отримати можливі варіанти шляху:

8.3. Пошук рішення через побудову дерев рішень

Простий приклад обходу дерева на Prolog може містити визначення структури дерева та рекурсивного обходу його вузлів. Розглянемо приклад із бінарним деревом.
% Визначення структури бінарного дерева
% Вузол дерева може бути або листом (порожнім), або мати значення та два піддерева (ліве та праве).
node(nil).
node(value(_, Left, Right)).

Розглянемо два основних методи обходу дерев: Пошук в глибину та Пошук в ширину. 
Пошук в глибину зазвичай організується за допомогою структури даних, яка називається стек – першим прийшов – останнім був оброблений. 

% Приклад бінарного дерева
sample_tree(
    value(1,
        value(2,
            value(4, nil, nil),
            value(5, nil, nil)
        ),
        value(3,
            value(6, nil, nil),
            value(7, nil, nil)
        )
    )
).


% Обхід дерева в глибину за допомогою робочого стеку
% Обхід: корінь -> ліве піддерево -> праве піддерево
preorder_stack(Tree) :-
    preorder_stack([Tree], []).

preorder_stack([], _).
preorder_stack([nil|Rest], Stack) :-
    preorder_stack(Rest, Stack).
preorder_stack([value(Value, Left, Right)|Rest], Stack) :-
    write(Value), write(' '),
    preorder_stack([Left, Right|Rest], Stack).

Видасть: 1 2 4 5 3 6 7
Цей код використовує два списки - один для стеку (Stack) і один для відвіданих вузлів (Rest). За допомогою цієї структури дані передаються між рекурсивними викликами, що дозволяє уникнути перевищення обмежень стеку.

Пошук в ширину зазвичай організується за допомогою структури даних, яка називається черга – першим прийшов – першим був оброблений. 

% Обхід дерева в ширину за допомогою черги
breadth_first(Tree) :-
    bfs([Tree]).

% Внутрішня функція для обходу в ширину
bfs([]).
bfs([nil|Rest]) :-
    bfs(Rest).
bfs([value(Value, Left, Right)|Rest]) :-
    write(Value), write(' '),
    enqueue(Left, Rest, NewRest),
    enqueue(Right, NewRest, FinalRest),
    bfs(FinalRest).

% Допоміжні предикати для керування чергою
enqueue(nil, Queue, Queue).
enqueue(Tree, Queue, NewQueue) :-
    append(Queue, [Tree], NewQueue).

У цьому коді, enqueue використовується для додавання елементів до черги. Перш ніж продовжити обробку, ліве та праве піддерева додаються до кінця черги. Потім вони витягуються з черги один за одним для обробки.
Результат: 1 2 3 4 5 6 7

Розглянемо просту експертну систему на базі обхода дерев.
Припустимо нам потрібно створити систему для рекомендації породи собак. 
Побудуємо деяке дерево рішень:

% Описуємо дерево рішень для вибору породи собак
dog_tree(
         if_then_else('Ви любите активні прогулянки та ігри на вулиці?',
             if_then_else('Вам подобаються великі собаки?',
                 recom('Порода: Лабрадор'),
                 recom('Порода: Бігль')
             ),
             if_then_else('Ви віддаєте перевагу компактним собакам?',
                 recom('Порода: Френч Бульдог'),
                 recom('Порода: Англійський бульдог')
             )
           )
         ).

% Виклик процедури обходу дерева
recommend_dog_breed :-
     dog_tree(Tree),
     tree_recom(Tree, Recommendation),
     write('Рекомендація:'), write(Recommendation), nl.

% Рекурсивна процедура обходу дерева
tree_recom(recom(A), A).
tree_recom(if_then_else(Question, Then, Else), A) :-
     is_true(Question, Response),
     (Response == так ->
         tree_recom(Then, A)
     ; tree_recom(Else, A)
     ).

% Діалог з користувачем
is_true(Question, Response) :-
     format("~w (так/ні)\n", [Question]),
     read(Response),
     member(Response, [так, ні]).

